TENSORS IN MAPLE - DEFINITION FROM TENSOR EQUATIONS

Link to: physicspages home page.

To leave a comment or report an error, please use the auxiliary blog and include the title or URL of this post in your comment.

Post date: 16 Mar 2021.

We've seen how to define tensors in Maple by specifying their rank. Here we'll have a look at defining tensors in terms of their components.

As usual, we start the session with

```
with(Physics):
Setup(coordinates = {X = [t, x, y, z]}, signature = `-+++`);
```

This defines a 4-dim Minkowski space with the signature $[t,x,y,z] = [-+++]$. The metric $g_{\mu\nu}$ is then given by

$$g_{\mu\nu} = \begin{bmatrix} -1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{1}$$

[A side note here on Maple's notation. When displaying tensor indices, Maple puts a comma between pairs of indices, so that $g_{\mu\nu}$ is displayed as $g_{\mu,\nu}$. However, when Maple converts expressions to Latex, the commas are omitted, so that the Latex displays correctly.]

We can define a tensor by specifying its components. Suppose we want to define the 4-momentum p, where the component p^0 is the energy and the other three components are the 3-momentum. We have

```
Define(p[mu] = [E, p__x, p__y, p__z]);
```

To define a symbol with a suffix in Maple, we type the symbol followed by 2 underscores and then the suffix. To see the covariant components we type

```
p[];
```

which displays

$$p_\mu = \begin{bmatrix} E & p_x & p_y & p_z \end{bmatrix} \tag{2}$$

To get the contravariant (upper index) form, we type

```
p[~];
```

with the result

$$p^{\mu} = \left[\begin{array}{cccc} -E & p_x & p_y & p_z \end{array} \right] \tag{3}$$

Note that the component $p^0 = -p_0$ which is correct for the metric 1.

In Minkowski space, Maple always interprets the time component as component 0. This is not always the first component; rather it is the component whose sign differs from the signs of the other three components. In our case, this happens to be the first component, but if we had defined the signature as

```
Setup(coordinates = {X = [t, x, y, z]}, signature = '+++-');
```

then the time component would be the last component.

We can access the time component by typing

```
p[0];
```

which displays E for p_0. Typing

```
p[~0];
```

displays $-E$ for p^0.

Now suppose we wish to define the Maxwell tensor $F_{\mu\nu}$ for electromagnetism. This tensor is defined in terms of the 4-potential A_μ by

$$F_{\mu\nu} = \partial_\mu A_\nu - \partial_\nu A_\mu \tag{4}$$

We can specify $F_{\mu\nu}$ as a 4×4 matrix:

```
F[mu, nu] = Matrix(4, symbol = F);
```

Maple responds with

$$F_{\mu\nu} = \left[\begin{array}{cccc} F_{11} & F_{12} & F_{13} & F_{14} \\ F_{21} & F_{22} & F_{23} & F_{24} \\ F_{31} & F_{32} & F_{33} & F_{34} \\ F_{41} & F_{42} & F_{43} & F_{44} \end{array} \right] \tag{5}$$

This statement on its own does *not* define the tensor $F_{\mu\nu}$ however. Defining a tensor *always* requires the use of the `Define()` statement. To actually create the tensor $F_{\mu\nu}$ we can now give the command

```
Define(<label>);
```

where `<label>` is the label number of 5 in the Maple worksheet. You can insert a label number into a command by typing `Ctrl-L`, which brings up a little dialog box into which you type the label number. You *cannot* just type the label number directly into the command, since Maple just interprets this as an ordinary number.

Having given the `Define` command, Maple responds with "Defined objects with tensor properties" and gives a list of the currently defined tensors.

$$\{A_\mu, \gamma_\mu, F_{\mu\nu}, \sigma_\mu, \partial_\mu, g_{\mu\nu}, p_\mu, \epsilon_{\alpha\beta\mu\nu}, X_\mu\} \tag{6}$$

We see that $F_{\mu\nu}$ is in the list, so it has now been defined. You can typ

```
F[];
```
to verify that the tensor has been defined. You will see the same output as in 5.

To see the upper-index version, give the command
```
F[~];
```
with the result

$$F^{\mu\nu} = \begin{bmatrix} F_{11} & -F_{12} & -F_{13} & -F_{14} \\ -F_{21} & F_{22} & F_{23} & F_{24} \\ -F_{31} & F_{32} & F_{33} & F_{34} \\ -F_{41} & F_{42} & F_{43} & F_{44} \end{bmatrix} \tag{7}$$

Note that all the components with a single time index (top row and first column) have now been made negative.

You can see mixed-index versions with commands like
```
F[~mu, nu, matrix];
```
which gives

$$F^{\mu}{}_{\nu} = \begin{bmatrix} -F_{11} & -F_{12} & -F_{13} & -F_{14} \\ F_{21} & F_{22} & F_{23} & F_{24} \\ F_{31} & F_{32} & F_{33} & F_{34} \\ F_{41} & F_{42} & F_{43} & F_{44} \end{bmatrix} \tag{8}$$

From its definition 4 we can see that $F_{\mu\nu} = -F_{\nu\mu}$ so the tensor is anti-symmetric. We can include this is the Maple definition with the command
```
Define(<label>,antisymmetric);
```
where <label> refers to the original command defining $F_{\mu\nu}$ following 4. We can see the result with the command
```
F[];
```
which displays

$$F_{\mu\nu} = \begin{bmatrix} 0 & F_{12} & F_{13} & F_{14} \\ -F_{12} & 0 & F_{23} & F_{24} \\ -F_{13} & -F_{23} & 0 & F_{34} \\ -F_{14} & -F_{24} & -F_{34} & 0 \end{bmatrix} \tag{9}$$

We still haven't included the definition 4, however. To do this we have
```
F[mu, nu] = 2*Antisymmetrize(d_[mu](A[nu](X)));
Define(redo,<label>);
```
The first line of this code defines $F_{\mu\nu}$ as twice the antisymmetrized version of $\partial_\mu A_\nu(x)$. We can antisymmetrize any rank-2 tensor $B_{\alpha\beta}$ by writing

$$B_{[\alpha\beta]} = \frac{1}{2}\left(B_{\alpha\beta} - B_{\beta\alpha}\right) \tag{10}$$

and this is the Antisymmetrize operation in Maple. The Define(redo,<label>) command tells Maple to redo the definition of $F_{\mu\nu}$ using the definition in the line with the number <label>, which in this case is the line immediately preceding. The notation d_[mu] indicates partial derivative, and to do a derivative, we need to specify the coordinates on which A_ν depends. That's the reason for the (X) in the definition. The X refers to the coordinate system we defined back at the start of this post, that is

$$X = [t, x, y, z] \tag{11}$$

To see the result, we type

```
F[];
```

with the result

$$F_{\mu\nu} = \left[\begin{array}{cccc} 0 & \frac{\partial}{\partial t}A_2(X) - \frac{\partial}{\partial x}A_1(X) & \frac{\partial}{\partial t}A_3(X) - \frac{\partial}{\partial y}A_1(X) & \frac{\partial}{\partial t}A_4(X) - \frac{\partial}{\partial z}A_1(X \\ \frac{\partial}{\partial x}A_1(X) - \frac{\partial}{\partial t}A_2(X) & 0 & \frac{\partial}{\partial x}A_3(X) - \frac{\partial}{\partial y}A_2(X) & \frac{\partial}{\partial x}A_4(X) - \frac{\partial}{\partial z}A_2(X \\ \frac{\partial}{\partial y}A_1(X) - \frac{\partial}{\partial t}A_3(X) & \frac{\partial}{\partial y}A_2(X) - \frac{\partial}{\partial x}A_3(X) & 0 & \frac{\partial}{\partial y}A_4(X) - \frac{\partial}{\partial z}A_3(X \\ \frac{\partial}{\partial z}A_1(X) - \frac{\partial}{\partial t}A_4(X) & \frac{\partial}{\partial z}A_2(X) - \frac{\partial}{\partial x}A_4(X) & \frac{\partial}{\partial z}A_3(X) - \frac{\partial}{\partial y}A_4(X) & 0 \end{array}\right.$$

The result won't fit on the page, but you get the idea. A more compact notation would be helpful. We can give the command

```
CompactDisplay(A(X));
```

to which Maple responds with "A(X) will now be displayed as A". We can now give the command

```
Define(redo, <label>);
```

Typing F[]; to see the result, we have

$$F_{\mu\nu} = \left[\begin{array}{cccc} 0 & (A_2)_t - (A_1)_x & (A_3)_t - (A_1)_y & (A_4)_t - (A_1)_z \\ (A_1)_x - (A_2)_t & 0 & (A_3)_x - (A_2)_y & (A_4)_x - (A_2)_z \\ (A_1)_y - (A_3)_t & (A_2)_y - (A_3)_x & 0 & (A_4)_y - (A_3)_z \\ (A_1)_z - (A_4)_t & (A_2)_z - (A_4)_x & (A_3)_z - (A_4)_y & 0 \end{array}\right] \tag{12}$$

The explicit (X) dependence has been dropped, and derivatives are now indicated with a suffix after enclosing the term to be differentiated within parentheses. This is the usual form for the Maxwell electromagnetic tensor.

We can see the number of independent components in $F_{\mu\nu}$ with the command

```
Library:-NumberOfIndependentTensorComponents(F);
```

which gives the result 6, which is correct. This command can be useful with more complex tensors such as the rank-4 Riemann curvature tensor.